Least Squares Fitting:
Physics 238

In Chapter 8 of *An Introduction to Error Analysis*, by John R. Taylor, the author discusses the general theme of least-squares fitting. This is based on the normal distribution discussion in Chapter 5. Here, we will take the theoretical background as given, and show how the minimization of the least square difference leads to the standard results for some simple cases.

In[1]:=
```
Clear["Global`*"]
```

Fitting a Straight Line

First, here is some sample data of {mass, angle} pairs from the torsional oscillator experiment. (See the separate importing notebook for more about importing data.

Data

In[2]:=
```
SetDirectory[NotebookDirectory[]]
```

Out[2]=
```
/Users/doughera/238/2025/lectures-dev/Ch08
```

In[3]:=
```
rawdata = Import["TO-static-20250212.csv", "CSV"]
```

Out[3]=
```
{{Mass (g), Raw Angle, Voltage}, {0, 2.98, -0.0456}, {50, 2.79, -0.4083},
 {100, 2.59, -0.7733}, {150, 2.38, -1.172}, {200, 2.17, -1.556},
 {250, 1.96, -1.958}, {300, 1.66, -2.483}, {350, 1.41, -2.726},
 {400, 1.21, -2.566}, {0, 2.98, -0.0606}, {-50, 3.2, 0.3049}, {-100, 3.4, 0.6583},
 {-150, 3.6, 1.028}, {-200, 3.81, 1.387}, {-250, 4.06, 1.812},
 {-300, 4.32, 2.237}, {-350, 4.56, 2.502}, {-400, 4.74, 2.369}, {0, 3., -0.04502}}
```

In[4]:=
```
data3 = Select[rawdata, NumberQ[#[[1]]] && NumberQ[#[[2]]] &]
```

Out[4]=
```
{{0, 2.98, -0.0456}, {50, 2.79, -0.4083}, {100, 2.59, -0.7733}, {150, 2.38, -1.172},
 {200, 2.17, -1.556}, {250, 1.96, -1.958}, {300, 1.66, -2.483},
 {350, 1.41, -2.726}, {400, 1.21, -2.566}, {0, 2.98, -0.0606}, {-50, 3.2, 0.3049},
 {-100, 3.4, 0.6583}, {-150, 3.6, 1.028}, {-200, 3.81, 1.387}, {-250, 4.06, 1.812},
 {-300, 4.32, 2.237}, {-350, 4.56, 2.502}, {-400, 4.74, 2.369}, {0, 3., -0.04502}}
```

For this data, we want columns 1 and 2.

In[5]:=
```
data = Table[{data3[[i, 1]] , data3[[i, 2]]}, {i, 1, Length[data3]}]
```

Out[5]=
```
{{0, 2.98}, {50, 2.79}, {100, 2.59}, {150, 2.38},
 {200, 2.17}, {250, 1.96}, {300, 1.66}, {350, 1.41}, {400, 1.21},
 {0, 2.98}, {-50, 3.2}, {-100, 3.4}, {-150, 3.6}, {-200, 3.81},
 {-250, 4.06}, {-300, 4.32}, {-350, 4.56}, {-400, 4.74}, {0, 3.}}
```

Plot the data.

Be sure to give your axes meaningful labels

In[6]:=
```
dataPlot = ListPlot[data,  ImageSize → Scaled[0.8], Frame → True,
  LabelStyle → Large, FrameLabel → {"Mass (g)", "Angle (radians)"}]
```

Out[6]=



Theoretical Curve: A straight-line fit.

Here we explore how to find the best-fit straight line by the method of least squares. First, define the target function. Use the unknown items (intercept a0, and slope a1) as parameters to the function.

In[7]:=
```
yfit[x_, a0_, a1_] := a0 + a1 x
```

Calculating χ^2

Define a χ^2 function for the linear fit. For given a0 and a1 values, compute the average of the differences squared between the data and the fit value. Within the fuction, the data is in {i, m} pairs, so we pull out the current of the i^{th} value with data[[i, 1]], and the mass of the i^{th} value with data[[i, 2]]. You can use the Classroom Assistant Pallete to format the sum, or you can use *Mathematica*'s Sum[ ] function directly. Use whichever is easier for you to read. The denominator has the '-2' because with N data

points and 2 free parameters, there are only N-2 degrees of freedom.

In[8]:=
```
calculateChisq[data_, a0_, a1_] :=
 1/(Length[data] - 2) Sum[(yfit[data[[i, 1]], a0, a1] - data[[i, 2]])^2, {i, 1, Length[data]}]
```

In[9]:=
```
calculateChisq[data_, a0_, a1_] :=
 Sum[(yfit[data[[i, 1]], a0, a1] - data[[i, 2]])^2, {i, 1, Length[data]}] /
  (Length[data] - 2)
```

Initial Explorations

This command builds an interactive window showing the data, the current fit, and the chi squared value (as the plot title). It draws vertical lines from each data point to the fit line. Move the a0 and a1 sliders to minimize χ^2. Mathematica's "Filling" option gets confused if the data isn't sorted, so let's go ahead and sort the data by the first entry in each line.

In[10]:=
```
data = SortBy[data, First]
```

Out[10]=
```
{{-400, 4.74}, {-350, 4.56}, {-300, 4.32}, {-250, 4.06}, {-200, 3.81}, {-150, 3.6},
 {-100, 3.4}, {-50, 3.2}, {0, 2.98}, {0, 2.98}, {0, 3.}, {50, 2.79}, {100, 2.59},
 {150, 2.38}, {200, 2.17}, {250, 1.96}, {300, 1.66}, {350, 1.41}, {400, 1.21}}
```

In[11]:=
```
LinearModelFit[data, {1, x}, x]
```

Out[11]=
```
FittedModel[ 2.99 - 0.00436 x ]
```

In[12]:=
```
Manipulate[
 fitted = Table[{data[[i, 1]], yfit[data[[i, 1]], a0, a1]}, {i, 1, Length[data]}];
 Show[{ListPlot[{data, fitted},
    Filling → {1 → {{2}, {Red, Black}}}, PlotLegends → {"data", "fitted"}],
   Plot[yfit[x, a0, a1], {x, -400, 400}]},
  PlotLabel → calculateChisq[data, a0, a1], ImageSize → Large] ,
 {{a0, 3}, 1, 4, 0.1,  Appearance → "Open"},
 {{a1, 0}, -0.01, 0.01, 0.001, Appearance → "Open"}
]
```

Out[12]=



Minimizing χ^2

Picking the intercept a0 = 0 (close to the expected value), look at how χ^2 varies with a1.

In[13]:=
```
Plot[calculateChisq[data, 0, a1 ], {a1, -0.01, 0}, AxesLabel → {"a1", "χ²"}]
```

Out[13]=



Picking a1 = --250 (again, close to the expected value), look at how χ^2 varies with a0.

In[14]:=
```
Plot[calculateChisq[data, a0, -250], {a0, 1, 5}, AxesLabel → {"a0", "χ²"}]
```

Out[14]=



Ultimately, the “best” fit involves a two-dimensional minimization of χ^2. Sometimes it helps to visualize this sort of thing as a 3D plot or a density plot.

In[15]:=
```
Plot3D[calculateChisq[data, a0, a1], {a0, 1, 5},
 {a1, -0.01, 0}, AxesLabel → {"a0", "a1", "χ²"}]
```

Out[15]=



In[16]:=
```
DensityPlot[calculateChisq[data, a0, a1], {a0, 1, 5}, {a1, -0.01, 0}]
```

Out[16]=



Mathematica can find the minimum here as well.

In[17]:=
```
result = FindMinimum[calculateChisq[data, a0, a1 ], {a0, a1}]
```

Out[17]=
```
{0.00135759, {a0 → 2.99053, a1 → -0.00436275}}
```

Finally, here is the best fit curve.

In[18]:=
```
Show[{
  dataPlot,
  Plot[yfit[x, a0, a1] /. result[[2]], {x, -400, 400}, PlotStyle → Red]
 }]
```

Out[18]=



Linear Model Fit

Mathematica can do this minimization automatically. The LinearModelFit[ ] function searches the "parameter space" for the minimum value of χ^2. It reports a "Confidence Interval" that reflects the curvature of χ^2 -- how much you can vary either parameter without making χ^2 too large.

One other interesting thing to note: Note how our χ^2 space has a "valley" (dark purple in the DensityPlot) where you can explore what happens if you change a0 but don't change a1, and vice-versa. This is reflected in the LinearModelFit[ ] report. Look at the "Correlation Matrix." It tells you, in essence, how well correlated each parameter is with the other. The 1's along the diagonal mean a0 is perfectly correlated with a0 (obviously) and a1 is perfectly correlated with a1. The off-diagonal elements tell you that a0 and a1 are mostly uncorrelated -- increasing one doesn't really affect the other. This isn't always true for fits. Sometimes, a change in one parameter can be partially compensated for by a change in another. In those cases, they are not completely independent parameters, and the Correla-

tion Matrix elements are further from zero.

In[19]:=
```
fit = LinearModelFit[data, x, x]
```

Out[19]=

FittedModel[2.99 − 0.00436 x]

See the on-line help for more information on dealing with the results from LinearModelFit. Here are some examples of things you can do with it.

In[20]:=
```
fit["BestFit"]
```

Out[20]=

2.99053 − 0.00436275 x

In[21]:=
```
fit["BestFitParameters"]
```

Out[21]=

{2.99053, −0.00436275}

In[22]:=
```
fit["ParameterConfidenceIntervalTable"]
```

Out[22]=

	Estimate	Standard Error	Confidence Interval
1	2.99053	0.00845294	{2.97269, 3.00836 }
x	−0.00436275	0.0000364825	{−0.00443972 , −0.00428577 }

You can extract the values from that table, though it's clunky.

In[55]:=
```
fit["ParameterConfidenceIntervalTableEntries"]
```

Out[55]=

```
{{2.99053, 0.00845294, {2.97269, 3.00836}},
 {-0.00436275, 0.0000364825, {-0.00443972, -0.00428577}}}
```

For example, here is the slope and its uncertainty (the slope is the second entry in the table):

In[56]:=
```
{slope, δslope} = fit["ParameterConfidenceIntervalTableEntries"][[2, {1, 2}]]
```

Out[56]=

{−0.00436275, 0.0000364825}

In[23]:=
```
fit["CorrelationMatrix"] // MatrixForm
```

Out[23]//MatrixForm=

$$\begin{pmatrix} 1. & -1.3594\times10^{-32} \\ -1.45388\times10^{-32} & 1. \end{pmatrix}$$

In[24]:=
```
Show[{ListPlot[data], Plot[fit[x] , {x, -400, 400}, PlotStyle → Red]},
 LabelStyle → Larger, AxesLabel → {"Angle (radians)", "Mass (g)"},
 ImageSize → Scaled[0.7]]
```

Out[24]=



In[25]:=
```
fit["EstimatedVariance"]
```

Out[25]=
```
0.00135759
```

This is the same as our χ^2.

In[26]:=
```
{a0, a1} = fit["BestFitParameters"]
```

Out[26]=
```
{2.99053, -0.00436275}
```

In[27]:=
```
calculateChisq[data, a0, a1]
```

Out[27]=
```
0.00135759
```

Interpreting the Uncertainties.

Mathematica gives you the uncertainties in the parameters, but you should also check whether the fit is reasonable. In particular, do the differences between the data points and the fit make sense? Is the size believable? Are there systematic trends?

Mathematica will report the "FitResiduals", which are the differences between each data point and the fitted curve. The average square of the residuals is called the "EstimatedVariance". (Actually you divide by N-2, since there are 2 degrees of freedom used up by the two fit parameters, a0 and a1.

In[28]:=
```
fit["FitResiduals"]
```

Out[28]=
```
{0.00437564, 0.0425129, 0.0206502, -0.0212126, -0.0530753, -0.0449381,
 -0.0268008, -0.00866357, -0.0105263, -0.0105263, 0.00947368, 0.0176109,
 0.0357482, 0.0438854, 0.0520227, 0.06016, -0.0217028, -0.0535655, -0.0354283}
```

In[57]:=
```
Total[fit["FitResiduals"]^2] / (Length[fit["FitResiduals"]] - 2)
```
Out[57]=
```
0.00135759
```

In[58]:=
```
fit["EstimatedVariance"]
```
Out[58]=
```
0.00135759
```

In[59]:=
```
Sqrt[fit["EstimatedVariance"]]
```
Out[59]=
```
0.0368455
```

The square root of the estimated variance is thus the RMS (root mean square) error -- the “typical” amount by which the fitted line misses the data. It has the same units as the original y values, and should be compared to the y uncertainties. Are differences of 0.04 radians significant? For this experiment, you can easily read the scale to 0.1 radian, so differences of 0.04 radians seem plausible.

Other Fit Properties.

This is a complete list of all available properties of the fit.

In[32]:=
```
fit["Properties"]
```
Out[32]=
```
{AdjustedRSquared, AIC, AICc, ANOVA, BasisFunctions, BetaDifferences,
 BestFit, BestFitParameters, BIC, CatcherMatrix, CoefficientOfVariation,
 CookDistances, CorrelationMatrix, CovarianceMatrix, CovarianceRatios, Data,
 Weights, DesignMatrix, DurbinWatsonD, Eigenstructure, EstimatedVariance,
 FitDifferences, FitResiduals, Function, FVarianceRatios, HatDiagonal,
 MeanPredictions, MeanPredictionBands, ParameterEstimates, PartialSumOfSquares,
 PredictedResponse, Properties, Response, RSquared, SequentialSumOfSquares,
 SingleDeletionVariances, SinglePredictions, SinglePredictionBands,
 StandardizedResiduals, StudentizedResiduals, VarianceInflationFactors}
```

In[33]:=
```
{"AdjustedRSquared", "AIC", "AICc", "ANOVA", "BasisFunctions", "BetaDifferences",
 "BestFit", "BestFitParameters", "BIC", "CatcherMatrix", "CoefficientOfVariation",
 "CookDistances", "CorrelationMatrix", "CovarianceMatrix", "CovarianceRatios",
 "Data", "Weights", "DesignMatrix", "DurbinWatsonD", "Eigenstructure",
 "EstimatedVariance", "FitDifferences", "FitResiduals", "Function",
 "FVarianceRatios", "HatDiagonal", "MeanPredictions", "MeanPredictionBands",
 "ParameterEstimates", "PartialSumOfSquares", "PredictedResponse",
 "Properties", "Response", "RSquared", "SequentialSumOfSquares",
 "SingleDeletionVariances", "SinglePredictions", "SinglePredictionBands",
 "StandardizedResiduals", "StudentizedResiduals", "VarianceInflationFactors"}
```

Out[33]=
```
{AdjustedRSquared, AIC, AICc, ANOVA, BasisFunctions, BetaDifferences,
 BestFit, BestFitParameters, BIC, CatcherMatrix, CoefficientOfVariation,
 CookDistances, CorrelationMatrix, CovarianceMatrix, CovarianceRatios, Data,
 Weights, DesignMatrix, DurbinWatsonD, Eigenstructure, EstimatedVariance,
 FitDifferences, FitResiduals, Function, FVarianceRatios, HatDiagonal,
 MeanPredictions, MeanPredictionBands, ParameterEstimates, PartialSumOfSquares,
 PredictedResponse, Properties, Response, RSquared, SequentialSumOfSquares,
 SingleDeletionVariances, SinglePredictions, SinglePredictionBands,
 StandardizedResiduals, StudentizedResiduals, VarianceInflationFactors}
```

A note about the “Rsquared” value

In[34]:=
```
fit["RSquared"]
```

Out[34]=
```
0.998813
```

This is close to 1.0. Does that mean you have a good fit? Not necessarily. Consider the following parabolic data set (with a small amount of random noise added)

In[75]:=
```
Histogram[RandomReal[{-1, 1}, 1000]]
```

Out[75]=



In[76]:=
```
parab = Table[{x, 1.0 + 0.5 x + 0.3 x^2 + RandomReal[{-1, 1}]}, {x, 1, 20, 2}]
```

Out[76]=
```
{{1, 0.957887}, {3, 5.13292}, {5, 11.3511}, {7, 18.3758}, {9, 28.9411},
 {11, 42.8917}, {13, 59.137}, {15, 75.6975}, {17, 95.6829}, {19, 119.043}}
```

Let's try fitting it with a straight line

In[88]:=
```
pfit = LinearModelFit[parab, {1, x}, x]
```

Out[88]=
```
FittedModel[ -19.6 + 6.53 x ]
```

In[80]:=
```
pfit["ParameterConfidenceIntervalTable"]
```

Out[80]=

	Estimate	Standard Error	Confidence Interval
1	−19.5693	6.16039	{−33.7752, −5.36345}
x	6.52905	0.534174	{5.29724, 7.76085}

In[81]:=
```
Show[{ListPlot[parab], Plot[pfit[x] , {x, 0, 20}, PlotStyle → Red]},
 LabelStyle → Larger, ImageSize → Scaled[0.6]]
```

Out[81]=



In[82]:=
```
pfit["RSquared"]
```

Out[82]=
```
0.949172
```

In[83]:=
```
Sqrt[pfit["EstimatedVariance"]]
```

Out[83]=
```
9.70375
```

This is still close to 1.0, even though we're missing a systematic trend in the data. The R^2 value does not distinguish between data points that scatter about a line and those that deviate systematically from the line. You need to look at the actual graph. Obviously, a parabola is the right fit here:

```
pfit2 = LinearModelFit[parab, {1, x, x^2}, x]
```

Out[84]=
```
FittedModel[ 0.399 + 0.568 x + 0.298 x^2 ]
```

In[87]:=
```
pfit2["ParameterConfidenceIntervalTable"]
```

Out[87]=

	Estimate	Standard Error	Confidence Interval
1	0.398767	0.625104	{−1.07937, 1.8769 }
x	0.568418	0.145148	{0.225198, 0.911637 }
x^2	0.298031	0.00703573	{0.281395 , 0.314668 }

In[85]:=
```
Sqrt[pfit2["EstimatedVariance"]]
```

Out[85]=
```
0.646675
```

In[86]:=
```
Show[{ListPlot[parab],
  Plot[{pfit[x], pfit2[x]} , {x, 0, 20}, PlotLegends → Automatic]},
 LabelStyle → Larger, ImageSize → Scaled[0.6]]
```

Out[86]=



Summary of Linear Fit

Read or input the data and transform into appropriate form.

In[42]:=
```
rawdata = Import["TO-static-20250212.csv", "CSV"];
```

In[43]:=
```
data3 = Select[rawdata, NumberQ[#[[1]]] && NumberQ[#[[2]]] &];
```

In[44]:=
```
data = Table[{data3[[i, 1]] , data3[[i, 2]]}, {i, 1, Length[data3]}];
```

In[45]:=
```
dataPlot = ListPlot[data,  ImageSize → Scaled[0.8], Frame → True,
   LabelStyle → Large, FrameLabel → {"Mass (g)", "Angle (radians)"}];
```

In[46]:=
```
fit = LinearModelFit[data, {1, m}, m]
```

Out[46]=
```
FittedModel[ 2.99 − 0.00436 m ]
```

In[47]:=
```
fit["BestFitParameters"]
```

Out[47]=
```
{2.99053, −0.00436275}
```

In[48]:=
```
fit["ParameterConfidenceIntervalTable"]
```

Out[48]=

	Estimate	Standard Error	Confidence Interval
1	2.99053	0.00845294	{2.97269, 3.00836 }
m	−0.00436275	0.0000364825	{−0.00443972 , −0.00428577 }

In[49]:=
```
Sqrt[fit["EstimatedVariance"]] (* Root Mean Square error *)
```

Out[49]=
```
0.0368455
```

Overlay plots

In[50]:=
```
Show[{
  dataPlot,
  Plot[fit[m], {m, -400, 400}, PlotStyle → Red]
 }, ImageSize → Scaled[0.5]
]
```

Out[50]=